

Programmation Fonctionnelle

Définition de types et de classes

Luigi Santocanale
LIF, Aix-Marseille Université
Marseille, FRANCE

4 octobre 2013

Plan

Nouveaux noms pour les types : type

Nouveaux types : data (et newtype)

Les types récurifs

Définition des classes

Exercices

Plan

Nouveaux noms pour les types : `type`

Nouveaux types : `data` (et `newtype`)

Les types récursifs

Définition des classes

Exercices

Les déclarations *type*

En Haskell, un nouveaux *nom* d'un type peut se définir par le mot clé `type`.

```
type String = [Char]
```

String est un synonyme pour le type [Char].

Les déclarations de type sont utilisées pour rendre les autres types (et surtout les types des fonctions) plus faciles à lire.

Par exemple, étant donnée la déclaration

```
type Position = (Int,Int)
```

nous pouvons écrire

```
origine :: Position  
origine = (0,0)
```

```
a_gauche :: Position -> Position  
a_gauche = (x,y) = (x-1,y)
```

D'autres exemples – des TPs

```
-- nombres complexes
type Complexe = (Float,Float)

-- images
type Pixel = Int
type Image = [[Pixel]]
type Voisinage = [Pixel]
type Effet = (Pixel,Voisinage) -> Pixel
appliquerEffet :: Effet -> Image -> Image

-- nim
type Echiquier = [Int]
```

Déclaration de types avec paramètres

Les déclarations de type peuvent avoir des paramètres.

Par exemple, étant donnée la déclaration

```
type Paire a = (a,a)
```

nous pouvons écrire

```
mult :: Paire Int -> Int  
mult (m,n) = m*n
```

```
copy :: a -> Paire a  
copy x = (x,x)
```

Les déclarations de type peuvent être entrelacées :

```
type Position = (Int,Int)
type Transition = Position -> Position
```

Par contre, elle ne peuvent pas être récursives :

```
type Tree = (Int, [Tree])
```

NO!!!

Plan

Nouveaux noms pour les types : `type`

Nouveaux types : `data` (et `newtype`)

Les types récursifs

Définition des classes

Exercices

Les déclarations *Data*

On définit *nouveau type* par le mot clé `data`.

Par exemple, on trouve dans `Prelude` la déclaration suivante :

```
data Bool = False | True
```

*Bool est un nouveau type,
avec deux nouvelles valeurs, False et True.*

Remarque(s) :

- Les deux valeurs `False` et `True` sont appelés *constructeurs* du type `Bool`.
- Les noms des types et des constructeurs débutent toujours par une majuscule.
- Cf. les grammaires hors-contexte, qui spécifient les expressions bien formées d'un langage.
De façon analogue, les déclarations `data` spécifient ceux qui sont les éléments du nouveau type.

newtype

- définit un nouveau type,
- avec un seul constructeur,
- implémentations dans le compilateur améliorée par rapport à `data`.

Caveat

Mot clé	Quoi de neuf ?	type déclaré
type	le nom	préexistant
data	le type	est crée

Les valeurs d'un nouveau type s'utilisent comme les valeurs des autres types.

Par exemple, étant donnée la déclaration

```
data Reponse = Oui | Non | Inconnu
```

nous pouvons définir les fonctions suivantes :

```
reponses :: [Reponse]
reponses = [Oui, Non, Inconnu]
```

```
flip :: Reponse -> Reponse
flip Oui = Non
flip Non = Oui
flip Inconnu = Inconnu
```

Les paramètres avec *data*

Les constructeurs dans une déclaration `data` peuvent avoir des paramètres.

Par exemple, étant donné la déclaration

```
data Shape = Circle Float | Rect Float Float
```

nous pouvons définir

```
square :: Float -> Shape
square n = Rect n n
```

```
area :: Shape -> Float
area (Circle r) = pi * r^2
area (Rect x y) = x * y
```

Remarque(s) :

- Shape a valeurs de la forme `Circle r` où `r` est un valeur flottant, et de la forme `Rect x y` où `x` et `y` sont des valeurs flottants.

- `Circle` and `Rect` peuvent se voir comme des fonctions qui construisent des valeurs de type `Shape` :

```
Circle :: Float -> Shape
```

```
Rect   :: Float -> Float -> Shape
```

Les types déclarés avec `data` peuvent aussi être paramétrés.

Par exemple, on trouve dans `Prelude` :

```
data Maybe a = Nothing | Just a
```

d'où :

```
safediv :: Int -> Int -> Maybe Int
safediv _ 0 = Nothing
safediv m n = Just (m `div` n)
```

```
safehead :: [a] -> Maybe a
safehead [] = Nothing
safehead xs = Just (head xs)
```

Le type `Maybe` offre une première façon de traiter les erreurs :
on retourne `Nothing` si erreur.

Plan

Nouveaux noms pour les types : `type`

Nouveaux types : `data` (et `newtype`)

Les types récursifs

Définition des classes

Exercices

Types récurrents

Les nouveaux types – déclarés par `data` – peuvent se définir en terme de soi-mêmes. On dit alors qu'il sont *récurrents*.

Exemple :

```
data Nat = Zero | Succ Nat
```

*Nat est un nouveau type, avec constructeurs
Zero :: Nat et Succ :: Nat -> Nat.*

Remarque(s) :

- Un valeur de type `Nat`
 - ▶ soit il est `Zero`,
 - ▶ soit il est de la forme `Succ n` où `n :: Nat`.

- C'est-à-dire, `Nat` contient la suite infinie de valeurs suivants :

```
Zero
Succ Zero
Succ (Succ Zero)
...
```

- Nous pouvons considérer les valeurs de type `Nat` comme les *nombre naturels* (les entiers positifs ou égaux à 0), où
 - ▶ `Zero` représente 0,
 - ▶ `Succ` représente la fonction successeur $\lambda x \rightarrow x + 1$.

- Par exemple, la valeur

`Succ (Succ (Succ Zero))`

représente le nombre naturel

$$1 + (1 + (1 + 0)) = 3$$

C'est facile – en utilisant la récursion – de convertir les valeurs de type `Nat` vers `Int`, et vice-versa :

```
nat2int :: Nat -> Int
nat2int Zero = 0
nat2int (Succ n) = 1 + nat2int n

int2nat :: Int -> Nat
int2nat 0 = Zero
int2nat n = Succ (int2nat (n - 1))
```

et ainsi les additionner :

```
add :: Nat -> Nat -> Nat
add m n = int2nat (nat2int m + nat2int n)
```

Par ailleurs, c'est plus intéressant (et efficace) de définir l'addition directement sur les nombres naturels par récursion :

```
add Zero n = n
add (Succ m) n = Succ (add m n)
```

Par exemple :

```
add (Succ (Succ Zero)) (Succ Zero)
  = Succ (add (Succ Zero) (Succ Zero))
  = Succ (Succ (add Zero (Succ Zero)))
  = Succ (Succ (Succ Zero))
```

Remarque(s) :

- La définition récursive de `add` corresponde aux lois

$$0 + n = n, \quad (1 + m) + n = 1 + (n + m).$$

- La définition de la fonction `add` utilise les deux **motifs** :

Zero Succ m

(à la gache du symbole d'égalité)


Expressions arithmétiques

On considère une forme élémentaire d'expressions, construites des entiers en utilisant l'addition et la multiplication.

Par exemple :

ou, en forme arborescente :

$$1 + 2 * 3$$



On définit le type des expressions – par récursion – comme suit :

```
data Expr = Val Int
          | Add Expr Expr
          | Mul Expr Expr
```

L'expression du transparent précédent est représentée par :

```
Add (Val 1) (Mul (Val 2) (Val 3))
```

C'est facile — en utilisant la récursion et le filtrage — définir des fonctions pour manipuler ces expressions.

Par exemple :

```
taille :: Expr -> Int
taille (Val n) = 1
taille (Add x y) = 1 + taille x + taille y
taille (Mul x y) = 1 + taille x + taille y
```

```
eval :: Expr -> Int
eval (Val n) = n
eval (Add x y) = eval x + eval y
eval (Mul x y) = eval x * eval y
```

Remarque(s) :

- Les trois constructeurs ont les types suivants :


```
Val  :: Int -> Expr
Add  :: Expr -> Expr -> Expr
Mul  :: Expr -> Expr -> Expr
```

- Plusieurs fonctions sur les expressions peuvent se définir en remplaçant les constructeurs par d'autres fonctions – en utilisant donc une forme de `fold` appropriée. Par exemple :

```
eval = fold id (+) (*)
taille = fold (\_ -> 1) (+) (+)
```

Arbres (de recherche) binaires

C'est souvent utile de stocker les données dans des arbres binaires :



Spécification d'arbre binaire :

- ou bien c'est une feuille étiquetée par un entier,
- sinon, c'est un noeud, avec
 - ▶ un fils à la gauche (un autre arbre binaire),
 - ▶ un entier comme étiquette,
 - ▶ un fils à la droite (un autre arbre binaire).

Le mécanisme de déclaration `data` permet d'implémenter tout de suite cette spécification :

```
data ArbreBinaire = Feuille Int
                  | Noeud ArbreBinaire Int ArbreBinaire
```

L'arbre du transparent précédent se représente alors comme suit :

```
Noeud (Noeud (Feuille 1) 3 (Feuille 4))
      5
      (Noeud (Feuille 6) 7 (Feuille 9))
```

C'est bien facile de définir une fonction qui décide si un entier a une occurrence dans un tel arbre :

```
occurs :: Int -> ArbreBinaire -> Bool
occurs m (Feuille n) = m==n
occurs m (Noeud l n r) = m==n
                        || occurs m l
                        || occurs m r
```

Mais ... dans le pire de cas – si l'entier n'a pas une telle occurrence – cette fonction parcourt l'arbre entier.

Considérez la fonction `flatten` qui transforme un arbre dans une liste, en lisant les étiquettes de l'arbre selon un parcours en profondeur gauche :

```
flatten :: ArbreBinaire -> [Int]
flatten (Feuille n) = [n]
flatten (Noeud l n r) =
    flatten l ++ [n] ++ flatten r
```

Un arbre binaire est un *arbre de recherche binaire* si la liste retournée par `flatten` est triée.

Par exemple, l'arbre de notre exemple est un arbre de recherche binaire, car la liste retournée est `[1,3,4,5,6,7,9]`.

Les arbres de recherche binaires ont la propriété importante que, quand on souhaite chercher un valeur, on peut toujours décider dans quel sous-arbre chercher :

```
occurs m (Feuille n) = m==n
occurs m (Noeud l n r)
  | m==n = True
  | m<n = occurs m l
  | m>n = occurs m r
```

Cette nouvelle définition est bien plus efficient, car elle parcours seulement une branche de l'arbre.

Si l'arbre est *balancé*, alors une branche a au plus $\log_2(n)$ éléments, où n est la taille de l'arbre.

Plan

Nouveaux noms pour les types : type

Nouveaux types : data (et newtype)

Les types récursifs

Définition des classes

Exercices

Déclaration d'une classe (*class*)

On trouve, dans Prelude, la déclaration suivante :

```
class Eq a where
  (==)  :: a -> a -> Bool
  (/=)  :: a -> a -> Bool
  x /= y = not (x == y)
```

La classe Eq a deux methodes :

- *l'égalité : ==*
- *l'inegalité : /=*

L'inégalité y est aussi définie – de façon standard – en fonction de l'égalité.

Instancier une classe (*instance*)

Bool est un membre de la classe Eq.

On trouve, dans Prelude, la déclaration suivante :

```
instance Eq Bool where
  False == False = True
  True  == True  = True
  _     == _     = False
```

Le type Bool appartient à la classe Eq :

- *on donne la définition de l'égalité,*
- *on donne pas la définition de l'inégalité,
– on utilisera alors la définition standard.*

Extension de classes

La classe `Ord` est une extension de la classe `Eq` :

```
class Eq a => Ord a where
  (<), (<=), (>), (>=) :: a -> a -> Bool
  min, max :: a -> a -> a
```

*En plus de `==` et `/=`, `Ord` a six autres methodes :
`<`, `<=`, `>`, `>=`, `min`, `max`*

Remarques :

- Seulement un nouveau type – déclaré via `data` – peut être une instance d'une classe.
- Dans la déclaration de `Ord`, on trouve les définitions suivantes de `min` et `max` :

```
class Eq a => Ord a where
  ...
  min x y
    | x <= y = x
    | otherwise = y
  max x y
    | x <= y = y
    | otherwise = x
```

Instanciation automatique (*deriving*)

Nous pouvons instancier un nouveau type à des classes prédéfinie, via le mot clés `deriving` :

```
data Reponse = Non | Inconnu | Oui
              deriving (Eq,Ord,Show,Read)
```

Par exemple :

```
*Main> Non < Inconnu
True
```

Définir un module

Par exemple :

```
module Complexe (  
    Complexe,  
    re,  
    im,  
    construct,  
    origin,  
    modulus,  
    angle  
)  
  
where  
  
newtype Complexe = Paire (Float,Float)  
    deriving (Eq)  
  
...
```

Plan

Nouveaux noms pour les types : type

Nouveaux types : data (et newtype)

Les types récursifs

Définition des classes

Exercices

Exercices I

1. Utilisez la récursion et la fonction `add` pour définir une fonction qui multiplie le nombre naturels.
2. Définissez une fonction `fold` adaptée aux expressions, et donnez quelques exemple d'utilisation.
3. Précisez ce que veut dire qu'un arbre binaire est balancé. Définissez une fonction qui décide si un arbre binaire est balancé.
4. Définissez le type de `Complexe` nombre complexes, et faites de lui une instance de la classe `Num`.
5. Définissez le type `Image` des images, et instanciez ce type à la classe `Show`.

6. Voici une possible déclaration de la classe Set :

```
class Set s where
  appartient :: Eq a => a -> s a -> Bool
  ajouter   :: Eq a => a -> s a -> s a
  enlever   :: Eq a => a -> s a -> s a
  pick_one  :: Eq a => s a -> a
  vide      :: s a
  est_vide  :: s a -> Bool
  union     :: Eq a => s a -> s a -> s a
  intersection :: Eq a => s a -> s a -> s a
  difference :: Eq a => s a -> s a -> s a
```

Proposez deux types – un basé sur les listes, l'autre sur les arbres binaires – et instanciez cette classe à ces types.